
Решение Задания 3
Курс «Поиск медленных запросов и 4 приема их оптимизации»

Оптимальный запрос 1:
Оптимизация заключается в вынесении виртуальной таблицы в подзапрос

ВЫБРАТЬ
ПродажаТЧ.Биржа КАК Биржа,
ПродажаТЧ.ЦеннаяБумага КАК ЦеннаяБумага,
СУММА(ПродажаТЧ.Количество) КАК КоличествоНаПродажу,
ПродажаТЧ.Ссылка КАК Ссылка

ПОМЕСТИТЬ ВТ_ТабЧасть
ИЗ

Документ.Продажа.ТЧ КАК ПродажаТЧ
ГДЕ

ПродажаТЧ.Ссылка = &Ссылка

СГРУППИРОВАТЬ ПО
ПродажаТЧ.Биржа,
ПродажаТЧ.ЦеннаяБумага,
ПродажаТЧ.Ссылка

;

////////////////////////////////////////////////////////////////////////////////
ВЫБРАТЬ

ВТ_ТабЧасть.Биржа КАК Биржа,
ВТ_ТабЧасть.ЦеннаяБумага КАК ЦеннаяБумага,
КотировкиСрезПоследних.Цена КАК ЦенаПродажи,
ВТ_ТабЧасть.КоличествоНаПродажу КАК КоличествоНаПродажу,
ВТ_ТабЧасть.КоличествоНаПродажу * КотировкиСрезПоследних.Цена КАК СуммаПродажи

ПОМЕСТИТЬ ВТ_Котировки
ИЗ

ВТ_ТабЧасть КАК ВТ_ТабЧасть
ВНУТРЕННЕЕ СОЕДИНЕНИЕ РегистрСведений.Котировки.СрезПоследних(

&ДатаДокумента,
(Биржа, ЦеннаяБумага) В

(ВЫБРАТЬ
ВТ.Биржа,
ВТ.ЦеннаяБумага

ИЗ
ВТ_ТабЧасть КАК ВТ)) КАК КотировкиСрезПоследних

ПО ВТ_ТабЧасть.Биржа = КотировкиСрезПоследних.Биржа
И ВТ_ТабЧасть.ЦеннаяБумага = КотировкиСрезПоследних.ЦеннаяБумага

;

////////////////////////////////////////////////////////////////////////////////
ВЫБРАТЬ

ВТ_Котировки.Биржа КАК Биржа,
ВТ_Котировки.ЦеннаяБумага КАК ЦеннаяБумага,
АктивыОстатки.КоличествоОстаток КАК КоличествоОстаток,
ВТ_Котировки.ЦенаПродажи КАК ЦенаПродажи,
ВТ_Котировки.КоличествоНаПродажу КАК КоличествоНаПродажу,
ВТ_Котировки.СуммаПродажи КАК СуммаПродажи,
АктивыОстатки.СуммаОстаток КАК СуммаОстаток,
АктивыОстатки.ДокументПокупка КАК ДокументПокупка

ИЗ
РегистрНакопления.Активы.Остатки(&ДатаДокумента, ) КАК АктивыОстатки

ВНУТРЕННЕЕ СОЕДИНЕНИЕ ВТ_Котировки КАК ВТ_Котировки
ПО АктивыОстатки.ЦеннаяБумага = ВТ_Котировки.ЦеннаяБумага

И АктивыОстатки.Биржа = ВТ_Котировки.Биржа
И (АктивыОстатки.Клиент = &Клиент)

УПОРЯДОЧИТЬ ПО
АктивыОстатки.ДокументПокупка.Дата,
Биржа,
ЦеннаяБумага,
КоличествоОстаток,
ЦенаПродажи

ИТОГИ
СУММА(КоличествоОстаток),
МАКСИМУМ(КоличествоНаПродажу),
МАКСИМУМ(СуммаПродажи),
СУММА(СуммаОстаток)

ПО
Биржа,
ЦеннаяБумага

Оптимальный запрос 2:
Оптимизация заключается том, чтобы избавиться от ВЫБОР КОГДА, это вполне можно заменить другим
логическим выражением.

ВЫБРАТЬ
Котировки.Период КАК Период,
Котировки.Биржа КАК Биржа,
Котировки.ЦеннаяБумага КАК ЦеннаяБумага,
Котировки.Цена КАК Цена

ИЗ
РегистрСведений.Котировки КАК Котировки

ГДЕ
(Котировки.ЦеннаяБумага = &ЦеннаяБумага1

И Котировки.Цена > 500000
ИЛИ Котировки.ЦеннаяБумага <> &ЦеннаяБумага1

И Котировки.Цена < 0.0002)

Оптимальный запрос 3:
Оптимизация заключается том, чтобы избавиться от ИЛИ между разными таблицами.

Вариант 1:

ВЫБРАТЬ
Активы.ЦеннаяБумага КАК ЦеннаяБумага,
СУММА(Активы.Количество) КАК Количество,
Активы.Регистратор КАК Регистратор

ИЗ
РегистрНакопления.Активы КАК Активы

ГДЕ
(Активы.Регистратор = &Ссылка

ИЛИ Активы.ДокументПокупка = &Ссылка)

СГРУППИРОВАТЬ ПО
Активы.ЦеннаяБумага,
Активы.Регистратор

УПОРЯДОЧИТЬ ПО
ЦеннаяБумага

Вариант 2:

ВЫБРАТЬ
ВложенныйЗапрос.ЦеннаяБумага КАК ЦеннаяБумага,
СУММА(ВложенныйЗапрос.Количество) КАК Количество,
ВложенныйЗапрос.Регистратор КАК Регистратор

ИЗ
(ВЫБРАТЬ

Активы.ЦеннаяБумага КАК ЦеннаяБумага,
СУММА(Активы.Количество) КАК Количество,
Активы.Регистратор КАК Регистратор

ИЗ
РегистрНакопления.Активы КАК Активы

ГДЕ
Активы.Регистратор = &Ссылка

СГРУППИРОВАТЬ ПО
Активы.ЦеннаяБумага,
Активы.Регистратор

ОБЪЕДИНИТЬ

ВЫБРАТЬ
Активы.ЦеннаяБумага,
СУММА(Активы.Количество),
Активы.Регистратор

ИЗ
РегистрНакопления.Активы КАК Активы

ГДЕ
Активы.ДокументПокупка = &Ссылка

СГРУППИРОВАТЬ ПО
Активы.ЦеннаяБумага,
Активы.Регистратор) КАК ВложенныйЗапрос

СГРУППИРОВАТЬ ПО
ВложенныйЗапрос.ЦеннаяБумага,
ВложенныйЗапрос.Регистратор

УПОРЯДОЧИТЬ ПО
ЦеннаяБумага

Оптимальный запрос 4:
Оптимизация заключается том, чтобы избавиться от функций над полями условий.

Вариант 1:

ВЫБРАТЬ
Котировки.Период КАК Период,
Котировки.Биржа КАК Биржа,
Котировки.ЦеннаяБумага КАК ЦеннаяБумага,
Котировки.Цена КАК Цена

ИЗ
РегистрСведений.Котировки КАК Котировки

ГДЕ
Котировки.Период МЕЖДУ ДАТАВРЕМЯ(2019, 12, 1) И ДАТАВРЕМЯ(2019, 12, 31, 23, 59, 59)
ИЛИ Котировки.Период МЕЖДУ ДАТАВРЕМЯ(2020, 12, 1) И ДАТАВРЕМЯ(2020, 12, 31, 23, 59, 59)

Вариант 2:

ВЫБРАТЬ
Котировки.Период КАК Период,
Котировки.Биржа КАК Биржа,
Котировки.ЦеннаяБумага КАК ЦеннаяБумага,
Котировки.Цена КАК Цена

ИЗ
РегистрСведений.Котировки КАК Котировки

ГДЕ
Котировки.Период МЕЖДУ ДАТАВРЕМЯ(2019, 12, 1) И ДАТАВРЕМЯ(2019, 12, 31, 23, 59, 59)

ОБЪЕДИНИТЬ

ВЫБРАТЬ
Котировки.Период,
Котировки.Биржа,
Котировки.ЦеннаяБумага,
Котировки.Цена

ИЗ
РегистрСведений.Котировки КАК Котировки

ГДЕ
Котировки.Период МЕЖДУ ДАТАВРЕМЯ(2020, 12, 1) И ДАТАВРЕМЯ(2020, 12, 31, 23, 59, 59)

